

Optimizing DiffEq Code

Chris Rackauckas

February 23, 2019

In this notebook we will walk through some of the main tools for optimizing your code in order to efficiently solve DifferentialEquations.jl. User-side optimizations are important because, for sufficiently difficult problems, most of the time will be spent inside of your `f` function, the function you are trying to solve. "Efficient" integrators are those that reduce the required number of `f` calls to hit the error tolerance. The main ideas for optimizing your DiffEq code, or any Julia function, are the following:

- Make it non-allocating
- Use StaticArrays for small arrays
- Use broadcast fusion
- Make it type-stable
- Reduce redundant calculations
- Make use of BLAS calls
- Optimize algorithm choice

We'll discuss these strategies in the context of small and large systems. Let's start with small systems.

0.1 Optimizing Small Systems (<100 DEs)

Let's take the classic Lorenz system from before. Let's start by naively writing the system in its out-of-place form:

```
function lorenz(u,p,t)
    dx = 10.0*(u[2]-u[1])
    dy = u[1]*(28.0-u[3]) - u[2]
    dz = u[1]*u[2] - (8/3)*u[3]
    [dx,dy,dz]
end
```

```
lorenz (generic function with 1 method)
```

Here, `lorenz` returns an object, `[dx,dy,dz]`, which is created within the body of `lorenz`.

This is a common code pattern from high-level languages like MATLAB, SciPy, or R's `deSolve`. However, the issue with this form is that it allocates a vector, `[dx,dy,dz]`, at each step. Let's benchmark the solution process with this choice of function:

```
using DifferentialEquations, BenchmarkTools
u0 = [1.0;0.0;0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz,u0,tspan)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 11.30 MiB
  allocs estimate: 126166
  -----
  minimum time:      3.839 ms (0.00% GC)
  median time:       6.410 ms (34.53% GC)
  mean time:         5.948 ms (26.95% GC)
  maximum time:      26.875 ms (57.01% GC)
  -----
  samples:            839
  evals/sample:       1
```

The BenchmarkTools package's `@benchmark` runs the code multiple times to get an accurate measurement. The minimum time is the time it takes when your OS and other background processes aren't getting in the way. Notice that in this case it takes about 5ms to solve and allocates around 11.11 MiB. However, if we were to use this inside of a real user code we'd see a lot of time spent doing garbage collection (GC) to clean up all of the arrays we made. Even if we turn off saving we have these allocations.

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate: 9.93 MiB
  allocs estimate: 113274
  -----
  minimum time:      3.377 ms (0.00% GC)
  median time:       5.633 ms (0.00% GC)
  mean time:         5.057 ms (19.53% GC)
  maximum time:      8.446 ms (31.40% GC)
  -----
  samples:            990
  evals/sample:       1
```

The problem of course is that arrays are created every time our derivative function is called. This function is called multiple times per step and is thus the main source of memory usage. To fix this, we can use the in-place form to `***make our code non-allocating***`:

```
function lorenz!(du,u,p,t)
  du[1] = 10.0*(u[2]-u[1])
```

```

du[2] = u[1]*(28.0-u[3]) - u[2]
du[3] = u[1]*u[2] - (8/3)*u[3]
end

```

```
lorenz! (generic function with 1 method)
```

Here, instead of creating an array each time, we utilized the cache array `du`. When the inplace form is used, `DifferentialEquations.jl` takes a different internal route that minimizes the internal allocations as well. When we benchmark this function, we will see quite a difference.

```

u0 = [1.0;0.0;0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz!,u0,tspan)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
 memory estimate:  1.86 MiB
 allocs estimate:  44799
-----
 minimum time:      1.213 ms (0.00% GC)
 median time:      1.275 ms (0.00% GC)
 mean time:        1.638 ms (15.89% GC)
 maximum time:     18.317 ms (88.29% GC)
-----
 samples:           3044
 evals/sample:      1

```

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```

BenchmarkTools.Trial:
 memory estimate:  504.30 KiB
 allocs estimate:  31927
-----
 minimum time:      869.599 μs (0.00% GC)
 median time:      901.600 μs (0.00% GC)
 mean time:        992.511 μs (4.07% GC)
 maximum time:     12.208 ms (88.98% GC)
-----
 samples:           5039
 evals/sample:      1

```

There is a 4x time difference just from that change! Notice there are still some allocations and this is due to the construction of the integration cache. But this doesn't scale with the problem size:

```

tspan = (0.0,500.0) # 5x longer than before
prob = ODEProblem(lorenz!,u0,tspan)
@benchmark solve(prob,Tsit5(),save_everystep=false)

```

```

BenchmarkTools.Trial:
  memory estimate:  2.51 MiB
  allocs estimate:  164407
  -----
  minimum time:     4.467 ms (0.00% GC)
  median time:      4.603 ms (0.00% GC)
  mean time:        5.112 ms (2.87% GC)
  maximum time:     10.772 ms (0.00% GC)
  -----
  samples:          978
  evals/sample:     1

```

since that's all just setup allocations.

But if the system is small we can optimize even more. Allocations are only expensive if they are "heap allocations". For a more in-depth definition of heap allocations, [there are a lot of sources online](#). But a good working definition is that heap allocations are variable-sized slabs of memory which have to be pointed to, and this pointer indirection costs time. Additionally, the heap has to be managed and the garbage controllers has to actively keep track of what's on the heap.

However, there's an alternative to heap allocations, known as stack allocations. The stack is statically-sized (known at compile time) and thus its accesses are quick. Additionally, the exact block of memory is known in advance by the compiler, and thus re-using the memory is cheap. This means that allocating on the stack has essentially no cost!

Arrays have to be heap allocated because their size (and thus the amount of memory they take up) is determined at runtime. But there are structures in Julia which are stack-allocated. **structs** for example are stack-allocated "value-type"s. **Tuples** are a stack-allocated collection. The most useful data structure for DiffEq though is the **StaticArray** from the package [StaticArrays.jl](#). These arrays have their length determined at compile-time. They are created using macros attached to normal array expressions, for example:

```

using StaticArrays
A = @SVector [2.0,3.0,5.0]

```

Notice that the 3 after **SVector** gives the size of the **SVector**. It cannot be changed. Additionally, **SVectors** are immutable, so we have to create a new **SVector** to change values. But remember, we don't have to worry about allocations because this data structure is stack-allocated. **SArrays** have a lot of extra optimizations as well: they have fast matrix multiplication, fast QR factorizations, etc. which directly make use of the information about the size of the array. Thus, when possible they should be used.

Unfortunately static arrays can only be used for sufficiently small arrays. After a certain size, they are forced to heap allocate after some instructions and their compile time balloons. Thus static arrays shouldn't be used if your system has more than 100 variables. Additionally, only the native Julia algorithms can fully utilize static arrays.

Let's *****optimize `lorenz` using static arrays*****. Note that in this case, we want to use the out-of-place allocating form, but this time we want to output a static array:

```

function lorenz_static(u,p,t)
    dx = 10.0*(u[2]-u[1])
    dy = u[1]*(28.0-u[3]) - u[2]
    dz = u[1]*u[2] - (8/3)*u[3]
    @SVector [dx,dy,dz]
end

```

lorenz_static (generic function with 1 method)

To make the solver internally use static arrays, we simply give it a static array as the initial condition:

```

u0 = @SVector [1.0,0.0,0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz_static,u0,tspan)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
  memory estimate:  1.30 MiB
  allocs estimate:  42156
  -----
  minimum time:     958.300 μs (0.00% GC)
  median time:      987.250 μs (0.00% GC)
  mean time:        1.183 ms (13.36% GC)
  maximum time:     4.670 ms (78.36% GC)
  -----
  samples:          4230
  evals/sample:     1

```

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```

BenchmarkTools.Trial:
  memory estimate:  869.97 KiB
  allocs estimate:  39567
  -----
  minimum time:     866.899 μs (0.00% GC)
  median time:      890.350 μs (0.00% GC)
  mean time:        1.046 ms (11.27% GC)
  maximum time:     5.314 ms (71.16% GC)
  -----
  samples:          4770
  evals/sample:     1

```

And that's pretty much all there is to it. With static arrays you don't have to worry about allocating, so use operations like `*` and don't worry about fusing operations (discussed in the next section). Do "the vectorized code" of R/MATLAB/Python and your code in this case will be fast, or directly use the numbers/values.

Exercise 1 Implement the out-of-place array, in-place array, and out-of-place static array forms for the [Henon-Heiles System](#) and time the results.

0.2 Optimizing Large Systems

0.2.1 Interlude: Managing Allocations with Broadcast Fusion

When your system is sufficiently large, or you have to make use of a non-native Julia algorithm, you have to make use of `Arrays`. In order to use arrays in the most efficient manner, you need to be careful about temporary allocations. Vectorized calculations naturally have plenty of temporary array allocations. This is because a vectorized calculation outputs a vector. Thus:

```
A = rand(1000,1000); B = rand(1000,1000); C = rand(1000,1000)
test(A,B,C) = A + B + C
@benchmark test(A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  3
  -----
  minimum time:     3.081 ms (0.00% GC)
  median time:      3.333 ms (0.00% GC)
  mean time:        4.253 ms (18.66% GC)
  maximum time:     8.228 ms (37.61% GC)
  -----
  samples:          1174
  evals/sample:     1
```

That expression `A + B + C` creates 2 arrays. It first creates one for the output of `A + B`, then uses that result array to `+` `C` to get the final result. 2 arrays! We don't want that! The first thing to do to fix this is to use broadcast fusion. [Broadcast fusion](#) puts expressions together. For example, instead of doing the `+` operations separately, if we were to add them all at the same time, then we would only have a single array that's created. For example:

```
test2(A,B,C) = map((a,b,c)->a+b+c,A,B,C)
@benchmark test2(A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  5
  -----
  minimum time:     3.684 ms (0.00% GC)
  median time:      3.958 ms (0.00% GC)
  mean time:        4.866 ms (16.19% GC)
  maximum time:    22.997 ms (79.84% GC)
  -----
  samples:          1027
  evals/sample:     1
```

Puts the whole expression into a single function call, and thus only one array is required to store output. This is the same as writing the loop:

```
function test3(A,B,C)
    D = similar(A)
    @inbounds for i in eachindex(A)
        D[i] = A[i] + B[i] + C[i]
    end
    D
end
@benchmark test3(A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  2
  -----
  minimum time:     3.089 ms (0.00% GC)
  median time:      3.586 ms (0.00% GC)
  mean time:        4.675 ms (19.35% GC)
  maximum time:     16.465 ms (35.47% GC)
  -----
  samples:          1068
  evals/sample:     1
```

However, Julia's broadcast is syntactic sugar for this. If multiple expressions have a `.`, then it will put those vectorized operations together. Thus:

```
test4(A,B,C) = A .+ B .+ C
@benchmark test4(A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  2
  -----
  minimum time:     3.064 ms (0.00% GC)
  median time:      3.411 ms (0.00% GC)
  mean time:        4.376 ms (18.97% GC)
  maximum time:     10.677 ms (50.65% GC)
  -----
  samples:          1139
  evals/sample:     1
```

is a version with only 1 array created (the output). Note that `.s` can be used with function calls as well:

```
sin.(A) .+ sin.(B)
```

```
1000×1000 Array{Float64,2}:
 1.29528  1.32085  0.930357  1.49325  ...  0.159983  1.07498  0.967468
 1.01522  1.52475  0.110723  0.246455  ...  0.718995  0.950219  1.19527
 1.15823  0.84007  0.639067  1.18875  ...  1.02981  0.308393  0.99169
 1.02329  0.728823  0.490583  0.476998  ...  0.804252  1.46409  1.58224
 0.109182 0.472426  1.09831  1.47238  ...  1.0168  0.668954  1.22444
 0.409496 1.56038  0.728331  0.74371  ...  0.763565  1.29133  1.20796
```

0.474081	0.525426	0.828437	0.977098	0.892544	1.18938	0.8956
0.421441	1.28732	0.628917	1.17841	0.809665	0.89046	1.05373
0.16572	0.917762	1.22973	1.27326	0.259891	1.4146	1.07899
0.715804	0.642881	0.946889	1.24028	0.256946	1.34607	0.377025
⋮				⋮		
1.62647	1.524	0.432265	0.210637	0.691797	0.219371	0.735749
0.79603	0.702926	1.44137	0.820489	1.14669	0.883112	0.60478
1.34887	1.33906	1.23072	0.415831	0.725874	0.415191	1.07883
1.26446	1.06011	1.15027	0.848104	0.760909	1.53058	1.12544
0.164113	1.21472	1.1802	0.873886	... 0.795409	0.528863	1.05069
0.82855	0.649103	0.577686	1.20886	1.33716	1.19505	1.473
1.10221	0.318793	1.37995	1.3023	1.23873	0.838701	0.467891
0.647238	0.0999073	0.679561	0.58186	0.807387	0.285769	0.85452
1.54193	1.04922	0.49453	1.12757	0.84509	1.51574	1.46573

Also, the `@.` macro applies a dot to every operator:

```
test5(A,B,C) = @. A + B + C #only one array allocated
@benchmark test5(A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  3
  -----
  minimum time:     3.098 ms (0.00% GC)
  median time:      3.425 ms (0.00% GC)
  mean time:        4.383 ms (18.88% GC)
  maximum time:     9.277 ms (59.61% GC)
  -----
  samples:          1138
  evals/sample:     1
```

Using these tools we can get rid of our intermediate array allocations for many vectorized function calls. But we are still allocating the output array. To get rid of that allocation, we can instead use mutation. Mutating broadcast is done via `.=`. For example, if we pre-allocate the output:

```
D = zeros(1000,1000);
```

Then we can keep re-using this cache for subsequent calculations. The mutating broadcasting form is:

```
test6!(D,A,B,C) = D .= A .+ B .+ C #only one array allocated
@benchmark test6!(D,A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate:  0 bytes
  allocs estimate:  0
  -----
  minimum time:     1.582 ms (0.00% GC)
```



```

median time:      1.717 ms (0.00% GC)
mean time:        1.797 ms (0.00% GC)
maximum time:     4.651 ms (0.00% GC)
-----
samples:          2770
evals/sample:     1

```

If we use `@.` before the `=`, then it will turn it into `.=`:

```

test7!(D,A,B,C) = @. D = A + B + C #only one array allocated
@benchmark test7!(D,A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  0 bytes
 allocs estimate:  0
-----
minimum time:      1.574 ms (0.00% GC)
median time:       1.706 ms (0.00% GC)
mean time:        1.786 ms (0.00% GC)
maximum time:     4.296 ms (0.00% GC)
-----
samples:           2785
evals/sample:      1

```

Notice that in this case, there is no "output", and instead the values inside of `D` are what are changed (like with the `DiffEq` inplace function). Many Julia functions have a mutating form which is denoted with a `!`. For example, the mutating form of the `map` is `map!`:

```

test8!(D,A,B,C) = map!((a,b,c)->a+b+c,D,A,B,C)
@benchmark test8!(D,A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  32 bytes
 allocs estimate:  1
-----
minimum time:      1.720 ms (0.00% GC)
median time:       2.177 ms (0.00% GC)
mean time:        2.444 ms (0.00% GC)
maximum time:     13.469 ms (0.00% GC)
-----
samples:           2035
evals/sample:      1

```

Some operations require using an alternate mutating form in order to be fast. For example, matrix multiplication via `*` allocates a temporary:

```

@benchmark A*B

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB

```

```

allocs estimate:  2
-----
minimum time:    12.252 ms (0.00% GC)
median time:     14.089 ms (0.00% GC)
mean time:       15.057 ms (6.92% GC)
maximum time:    28.723 ms (14.60% GC)
-----
samples:         332
evals/sample:    1

```

Instead, we can use the mutating form `mul!` into a cache array to avoid allocating the output:

```

using LinearAlgebra
@benchmark mul!(D,A,B) # same as D = A * B

```

```

BenchmarkTools.Trial:
 memory estimate:  0 bytes
 allocs estimate:  0
-----
minimum time:     11.500 ms (0.00% GC)
median time:      13.198 ms (0.00% GC)
mean time:        17.893 ms (0.00% GC)
maximum time:     99.615 ms (0.00% GC)
-----
samples:          280
evals/sample:     1

```

For repeated calculations this reduced allocation can stop GC cycles and thus lead to more efficient code. Additionally, *****we can fuse together higher level linear algebra operations using BLAS*****. The package [SugarBLAS.jl](#) makes it easy to write higher level operations like `alpha*B*A + beta*C` as mutating BLAS calls.

0.2.2 Example Optimization: Gierer-Meinhardt Reaction-Diffusion PDE Discretization

Let's optimize the solution of a Reaction-Diffusion PDE's discretization. In its discretized form, this is the ODE:

$$du = D_1(A_y u + u A_x) + \frac{au^2}{v} + \bar{u} - \alpha u \quad (1)$$

$$dv = D_2(A_y v + v A_x) + au^2 + \beta v \quad (2)$$

where u , v , and A are matrices. Here, we will use the simplified version where A is the tridiagonal stencil $[1, -2, 1]$, i.e. it's the 2D discretization of the Laplacian. The native code would be something along the lines of:

```

# Generate the constants
p = (1.0,1.0,1.0,10.0,0.001,100.0) # a,α,ubar,β,D1,D2
N = 100

```

```

Ax = Array{Tridiagonal}([1.0 for i in 1:N-1],[-2.0 for i in 1:N],[1.0 for i in 1:N-1]))
Ay = copy(Ax)
Ax[2,1] = 2.0
Ax[end-1,end] = 2.0
Ay[1,2] = 2.0
Ay[end,end-1] = 2.0

function basic_version!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = r[:, :, 1]
    v = r[:, :, 2]
    Du = D1*(Ay*u + u*Ax)
    Dv = D2*(Ay*v + v*Ax)
    dr[:, :, 1] = Du .+ a.*u.*u./v .+ ubar .- α*u
    dr[:, :, 2] = Dv .+ a.*u.*u .- β*v
end

a,α,ubar,β,D1,D2 = p
uss = (ubar+β)/α
vss = (a/β)*uss^2
r0 = zeros{Float64}(100,100,2)
r0[:, :, 1] .= uss.+0.1.*rand.()
r0[:, :, 2] .= vss

prob = ODEProblem(basic_version!,r0,(0.0,0.1),p)

```

In this version we have encoded our initial condition to be a 3-dimensional array, with `u[:, :, 1]` being the A part and `u[:, :, 2]` being the B part.

```
@benchmark solve(prob,Tsit5())
```

```

BenchmarkTools.Trial:
 memory estimate: 186.89 MiB
 allocs estimate: 9176
-----
 minimum time:      72.694 ms (13.57% GC)
 median time:      152.969 ms (56.13% GC)
 mean time:        136.575 ms (47.71% GC)
 maximum time:     209.366 ms (61.43% GC)
-----
 samples:          37
 evals/sample:     1

```

While this version isn't very efficient,

We recommend writing the "high-level" code first, and iteratively optimizing it! The first thing that we can do is get rid of the slicing allocations. The operation `r[:, :, 1]` creates a temporary array instead of a "view", i.e. a pointer to the already existing memory. To make it a view, add `@view`. Note that we have to be careful with views because they point to the same memory, and thus changing a view changes the original values:

```

A = rand(4)
@show A

```

```
A = [0.690429, 0.668176, 0.359077, 0.315443]
```

```
B = @view A[1:3]
B[2] = 2
@show A
```

```
A = [0.690429, 2.0, 0.359077, 0.315443]
4-element Array{Float64,1}:
 0.6904285166672113
 2.0
 0.35907698242263386
 0.31544327118907156
```

Notice that changing B changed A. This is something to be careful of, but at the same time we want to use this since we want to modify the output `dr`. Additionally, the last statement is a purely element-wise operation, and thus we can make use of broadcast fusion there. Let's rewrite `basic_version!` to `***avoid slicing allocations***` and to `***use broadcast fusion***`:

```
function gm2!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = @view r[:,1]
    v = @view r[:,2]
    du = @view dr[:,1]
    dv = @view dr[:,2]
    Du = D1*(Ay*u + u*Ax)
    Dv = D2*(Ay*v + v*Ax)
    @. du = Du + a.*u.*u./v + ubar - α*u
    @. dv = Dv + a.*u.*u - β*v
end
prob = ODEProblem(gm2!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
 memory estimate: 119.57 MiB
  allocs estimate: 7706
  -----
 minimum time:      61.070 ms (10.13% GC)
 median time:      93.605 ms (32.46% GC)
 mean time:       114.736 ms (47.68% GC)
 maximum time:     181.700 ms (64.60% GC)
  -----
 samples:          45
 evals/sample:     1
```

Now, most of the allocations are taking place in `Du = D1*(Ay*u + u*Ax)` since those operations are vectorized and not mutating. We should instead replace the matrix multiplications with `mul!`. When doing so, we will need to have cache variables to write into. This looks like:

```

Ayu = zeros(N,N)
uAx = zeros(N,N)
Du = zeros(N,N)
Ayv = zeros(N,N)
vAx = zeros(N,N)
Dv = zeros(N,N)
function gm3!(dr,r,p,t)
    a, $\alpha$ ,ubar, $\beta$ ,D1,D2 = p
    u = @view r[:, :, 1]
    v = @view r[:, :, 2]
    du = @view dr[:, :, 1]
    dv = @view dr[:, :, 2]
    mul!(Ayu,Ay,u)
    mul!(uAx,u,Ax)
    mul!(Ayv,Ay,v)
    mul!(vAx,v,Ax)
    @. Du = D1*(Ayu + uAx)
    @. Dv = D2*(Ayv + vAx)
    @. du = Du + a*u*u./v + ubar -  $\alpha$ *u
    @. dv = Dv + a*u*u -  $\beta$ *v
end
prob = ODEProblem(gm3!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
 memory estimate: 29.77 MiB
 allocs estimate: 5942
-----
 minimum time:      48.138 ms (2.56% GC)
 median time:      49.398 ms (2.39% GC)
 mean time:        49.589 ms (3.01% GC)
 maximum time:     57.123 ms (1.78% GC)
-----
 samples:           101
 evals/sample:      1

```

But our temporary variables are global variables. We need to either declare the caches as `const` or localize them. We can localize them by adding them to the parameters, `p`. It's easier for the compiler to reason about local variables than global variables. *****Localizing variables helps to ensure type stability*****.

```

p = (1.0,1.0,1.0,10.0,0.001,100.0,Ayu,uAx,Du,Ayv,vAx,Dv) # a, $\alpha$ ,ubar, $\beta$ ,D1,D2
function gm4!(dr,r,p,t)
    a, $\alpha$ ,ubar, $\beta$ ,D1,D2,Ayu,uAx,Du,Ayv,vAx,Dv = p
    u = @view r[:, :, 1]
    v = @view r[:, :, 2]
    du = @view dr[:, :, 1]
    dv = @view dr[:, :, 2]
    mul!(Ayu,Ay,u)
    mul!(uAx,u,Ax)
    mul!(Ayv,Ay,v)
    mul!(vAx,v,Ax)
    @. Du = D1*(Ayu + uAx)
    @. Dv = D2*(Ayv + vAx)
    @. du = Du + a*u*u./v + ubar -  $\alpha$ *u
    @. dv = Dv + a*u*u -  $\beta$ *v
end

```

```

end
prob = ODEProblem(gm4!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
 memory estimate: 29.67 MiB
 allocs estimate: 1530
-----
 minimum time:      42.555 ms (2.63% GC)
 median time:      43.724 ms (2.62% GC)
 mean time:        44.036 ms (3.30% GC)
 maximum time:     54.884 ms (10.58% GC)
-----
 samples:          114
 evals/sample:     1

```

We could then use the BLAS `gemmv` to optimize the matrix multiplications some more, but instead let's devectorize the stencil.

```

p = (1.0,1.0,1.0,10.0,0.001,100.0,N)
function fast_gm!(du,u,p,t)
    a,α,ubar,β,D1,D2,N = p

    @inbounds for j in 2:N-1, i in 2:N-1
        du[i,j,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
            a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    end

    @inbounds for j in 2:N-1, i in 2:N-1
        du[i,j,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
            a*u[i,j,1]^2 - β*u[i,j,2]
    end

    @inbounds for j in 2:N-1
        i = 1
        du[1,j,1] = D1*(2u[i+1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
            a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    end

    @inbounds for j in 2:N-1
        i = 1
        du[1,j,2] = D2*(2u[i+1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
            a*u[i,j,1]^2 - β*u[i,j,2]
    end

    @inbounds for j in 2:N-1
        i = N
        du[end,j,1] = D1*(2u[i-1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
            a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    end

    @inbounds for j in 2:N-1
        i = N
        du[end,j,2] = D2*(2u[i-1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
            a*u[i,j,1]^2 - β*u[i,j,2]
    end

    @inbounds for i in 2:N-1
        j = 1
    end

```

```

    du[i,1,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
end
@inbounds for i in 2:N-1
    j = 1
    du[i,1,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end
@inbounds for i in 2:N-1
    j = N
    du[i,end,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
end
@inbounds for i in 2:N-1
    j = N
    du[i,end,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end

@inbounds begin
    i = 1; j = 1
    du[1,1,1] = D1*(2u[i+1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[1,1,2] = D2*(2u[i+1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = 1; j = N
    du[1,N,1] = D1*(2u[i+1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[1,N,2] = D2*(2u[i+1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = N; j = 1
    du[N,1,1] = D1*(2u[i-1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[N,1,2] = D2*(2u[i-1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = N; j = N
    du[end,end,1] = D1*(2u[i-1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[end,end,2] = D2*(2u[i-1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end
end
prob = ODEProblem(fast_gm!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

```

BenchmarkTools.Trial:

```

memory estimate: 29.64 MiB
allocs estimate: 1239
-----
minimum time:      8.878 ms (8.14% GC)
median time:       9.695 ms (9.57% GC)
mean time:         9.831 ms (11.18% GC)
maximum time:      19.600 ms (28.05% GC)
-----
samples:           508

```

```
evals/sample:      1
```

Lastly, we can do other things like multithread the main loops, but these optimizations get the last 2x-3x out. The main optimizations which apply everywhere are the ones we just performed (though the last one only works if your matrix is a stencil. This is known as a matrix-free implementation of the PDE discretization).

This gets us to about 8x faster than our original MATLAB/SciPy/R vectorized style code!

The last thing to do is then `***optimize our algorithm choice***`. We have been using `Tsit5()` as our test algorithm, but in reality this problem is a stiff PDE discretization and thus one recommendation is to use `CVODE_BDF()`. However, instead of using the default dense Jacobian, we should make use of the sparse Jacobian afforded by the problem. The Jacobian is the matrix $\frac{df_i}{dr_j}$, where r is read by the linear index (i.e. down columns). But since the u variables depend on the v , the band size here is large, and thus this will not do well with a Banded Jacobian solver. Instead, we utilize sparse Jacobian algorithms. `CVODE_BDF` allows us to use a sparse Newton-Krylov solver by setting `linear_solver = :GMRES` (see [the solver documentation](#), and thus we can solve this problem efficiently. Let's see how this scales as we increase the integration time.

```
prob = ODEProblem(fast_gm!,r0,(0.0,10.0),p)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate:  2.76 GiB
  allocs estimate:  111033
  -----
  minimum time:     2.263 s (19.44% GC)
  median time:      6.552 s (17.14% GC)
  mean time:        6.552 s (17.14% GC)
  maximum time:     10.841 s (16.66% GC)
  -----
  samples:          2
  evals/sample:     1
```

```
using Sundials
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES))
```

```
BenchmarkTools.Trial:
  memory estimate:  124.28 MiB
  allocs estimate:  37347
  -----
  minimum time:     569.203 ms (1.05% GC)
  median time:      682.264 ms (15.25% GC)
  mean time:        712.900 ms (14.46% GC)
  maximum time:     982.865 ms (18.88% GC)
  -----
  samples:          8
  evals/sample:     1
```



```

prob = ODEProblem(fast_gm!,r0,(0.0,100.0),p)
# Will go out of memory if we don't turn off `save_everystep`!
@benchmark solve(prob,Tsit5(),save_everystep=false)

```

```

BenchmarkTools.Trial:
 memory estimate: 19.82 MiB
 allocs estimate: 693067
-----
 minimum time:      4.906 s (0.06% GC)
 median time:      4.932 s (0.03% GC)
 mean time:        4.932 s (0.03% GC)
 maximum time:      4.957 s (0.00% GC)
-----
 samples:           2
 evals/sample:      1

```

```

@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES))

```

```

BenchmarkTools.Trial:
 memory estimate: 310.98 MiB
 allocs estimate: 94530
-----
 minimum time:      1.654 s (6.21% GC)
 median time:      1.707 s (6.02% GC)
 mean time:        1.803 s (7.84% GC)
 maximum time:      2.048 s (15.69% GC)
-----
 samples:           3
 evals/sample:      1

```

Now let's check the allocation growth.

```

@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES),save_everystep=false)

```

```

BenchmarkTools.Trial:
 memory estimate: 4.19 MiB
 allocs estimate: 85129
-----
 minimum time:      1.484 s (0.00% GC)
 median time:      1.489 s (0.00% GC)
 mean time:        1.493 s (0.00% GC)
 maximum time:      1.511 s (0.00% GC)
-----
 samples:           4
 evals/sample:      1

```

```

prob = ODEProblem(fast_gm!,r0,(0.0,500.0),p)
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES),save_everystep=false)

```

```

BenchmarkTools.Trial:
  memory estimate:  5.44 MiB
  allocs estimate: 117866
  -----
  minimum time:     2.057 s (0.00% GC)
  median time:      2.061 s (0.00% GC)
  mean time:        2.178 s (0.00% GC)
  maximum time:     2.416 s (0.00% GC)
  -----
  samples:          3
  evals/sample:     1

```

Notice that we've eliminated almost all allocations, allowing the code to grow without hitting garbage collection and slowing down.

Why is `CVODE_BDF` doing well? What's happening is that, because the problem is stiff, the number of steps required by the explicit Runge-Kutta method grows rapidly, whereas `CVODE_BDF` is taking large steps. Additionally, the `GMRES` linear solver form is quite an efficient way to solve the implicit system in this case. This is problem-dependent, and in many cases using a Krylov method effectively requires a preconditioner, so you need to play around with testing other algorithms and linear solvers to find out what works best with your problem.

0.3 Conclusion

Julia gives you the tools to optimize the solver "all the way", but you need to make use of it. The main thing to avoid is temporary allocations. For small systems, this is effectively done via static arrays. For large systems, this is done via in-place operations and cache arrays. Either way, the resulting solution can be immensely sped up over vectorized formulations by using these principles.

0.4 Appendix

```

using DiffEqTutorials
DiffEqTutorials.tutorial_footer(WEAVE_ARGS[:folder],WEAVE_ARGS[:file])

```

These benchmarks are part of the `DiffEqTutorials.jl` repository, found at:

<https://github.com/JuliaDiffEq/DiffEqTutorials.jl>

To locally run this tutorial, do the following commands:

```

using DiffEqTutorials
DiffEqTutorials.weave_file(".", "introduction/optimizing_diffeq_code.jmd")

```

Computer Information:

```

Julia Version 1.1.0
Commit 80516ca202 (2019-01-21 21:24 UTC)
Platform Info:
  OS: Windows (x86_64-w64-mingw32)

```

```

CPU: Intel(R) Core(TM) i7-8700 CPU @ 3.20GHz
WORD_SIZE: 64
LIBM: libopenlibm
LLVM: libLLVM-6.0.1 (ORCJIT, skylake)
Environment:
  JULIA_EDITOR = "C:\Users\accou\AppData\Local\atom\app-1.34.0\atom.exe" -a
  JULIA_NUM_THREADS = 6

```

Package Information:

```

  Status `C:\Users\accou\.julia\environments\v1.1\Project.toml`
  [c52e3926] Atom v0.7.14
  [6e4b80f9] BenchmarkTools v0.4.2
  [336ed68f] CSV v0.4.3
  [be33ccc6] CUDAnative v1.0.1
  [3a865a2d] CuArrays v0.9.1
  [a93c6f00] DataFrames v0.17.1
  [39dd38d3] Dierckx v0.4.1
  [aae7a2af] DiffEqFlux v0.2.0
  [c894b116] DiffEqJump v6.1.0+ [C:\Users\accou\.julia\dev\DiffEqJump`]
  [1130ab10] DiffEqParamEstim v1.5.1
  [225cb15b] DiffEqTutorials v0.0.0 [C:\Users\accou\.julia\external\DiffEq
Tutorials.jl`]
  [0c46a032] DifferentialEquations v6.3.0
  [587475ba] Flux v0.7.3
  [f6369f11] ForwardDiff v0.10.3+ [C:\Users\accou\.julia\dev\ForwardDiff`]
  [7073ff75] IJulia v1.17.0
  [c601a237] Interact v0.9.1
  [b6b21f68] Ipopt v0.5.4
  [4076af6c] JuMP v0.18.5
  [e5e0dc1b] Juno v0.5.4
  [76087f3c] NLOpt v0.5.1
  [429524aa] Optim v0.17.2
  [1dea7af3] OrdinaryDiffEq v5.1.4+ [C:\Users\accou\.julia\dev\OrdinaryDif
fEq`]
  [65888b18] ParameterizedFunctions v4.1.0
  [91a5bcdd] Plots v0.23.0
  [71ad9d73] PuMaS v0.0.0 [C:\Users\accou\.julia\dev\PuMaS`]
  [731186ca] RecursiveArrayTools v0.20.0
  [90137ffa] StaticArrays v0.10.2
  [789caeaf] StochasticDiffEq v6.1.1+ [C:\Users\accou\.julia\dev\Stochasti
cDiffEq`]
  [c3572dad] Sundials v3.0.0
  [44d3d7a6] Weave v0.7.1

```